

Trimix by blending Heliox and Air, on a HP49G / HP48 calculator
By partial pressure method, (Dalton law) using ideal gas and metric units
Release 1.4 1999-11-25 by Josep Guarro jguarro@retemail.es

This program is just a tool to help Trimix divers to plan and blend their bottom gas, by topping Heliox (or pure Helium) with Air.

The first thing is to decide wich Equivalent Narcotic Depth (END) you wish on your bottom mix at the maximum planned depth (MPD) of your dive, that is deffined as the depth on air at wich the inspired ppN2 has the same value that your Trimix at the maximum planned depth.

Usually when you plan a Trimix dive you choose an END in the 30 to 40m range to avoid inert gas narcosis. It's up to you to provide meaningfull values to the program.

Second thing is to enter the maximum planned depth (MPD) at wich you want to have the specified END, be aware that this depth can result greater than the maximum operative depth (MOD) of the Trimix, depending on wich Heliox premix you have, if that occurs you will see at the end of the program that the MOD is greater than the maximum planned detph you have entered, then you should have to use less Oxygen in your Heliox premix. Again is up to you to provide meaninfull data, of course if END is greater than maximum depth, the final results are meaningless.

Third thing is to enter the percentage of Oxygen in your Heliox, usually it should be from 5 to 25% depending on the max depth you are going to reach.

Last thing is to provide the final pressure you want in your dive tanks, so the program calculates the amount of heliox you need to transfer to your tanks before topping them with air.

Example dive:

Equiv. Narc.Depth(m) ?	35
Max.Planned Depth(m)?	75
%O2 in Heliox premix ?	10
Final Pressure (bar) ?	230

Will give you:

```
Trimix: 16./42.  
O2: 15.8%  
He: 42.4%  
ppO2: 1.35 bar  
MOD: 78. m  
Heliox: 108. bar  
Air: 122. bar  
V8 Trimix
```

Do not forget to analize the Oxygen contents in your final mix before you dive!

How the program works :

You need to decide the four parameters, **END**, **MPD**, **O2HE** and **Pf**, remember:

- 1) **END** is the Equivalent Narcotic Depth you want at the maximum depth of the dive in meters
- 2) **MPD** is the Maximum Planned Depth of the dive in meters
- 3) **O2HE** is the % O₂ content in your Heliox premix, if you are blending HeliAir just set **O2HE** to zero
- 4) **Pf** is the final pressure you want in your dive tanks in bars

Given that four parameters the program calculates the following output, %O₂, %He, ppO₂, MOD_{1.4}, ppHeOx, ppAir, calculated as follows:

- 1) %O₂ is the % Oxygen contents in the final Trimix

$$\%O_2 = \frac{21 \cdot (\text{END} + 10) + O2HE \cdot (\text{MPD} - \text{END})}{\text{MPD} + 10}$$

- 2) %He is the % Helium contents in the final Trimix

$$\%He = \frac{(\text{MPD} - \text{END}) \cdot (100 - O2HE)}{\text{MPD} + 10}$$

- 3) ppO₂ is the inspired Oxygen partial pressure, at the maximum planned depth of the dive in bars

$$ppO_2 = \frac{21 \cdot (\text{END} + 10) + O2HE \cdot (\text{MPD} - \text{END})}{1000}$$

- 4) MOD_{1.4} is the Maximum Operative Depth in meters of the Trimix accounting for a max ppO₂ of 1.4 bar

$$MOD_{1.4} = \frac{1400 \cdot (\text{MPD} + 10)}{21 \cdot (\text{END} + 10) + O2HE \cdot (\text{MPD} - \text{END})} + 10$$

- 5) ppHeOx is the Heliox premix pressure in bar you need to transfer to your dive tanks

$$ppHeOx = \frac{Pf \cdot (\text{MPD} - \text{END})}{\text{MPD} + 10}$$

- 6) ppAir is the Air pressure in bar you need to top in your dive tanks after the Heliox to get the final Trimix

$$ppAir = Pf - \frac{Pf \cdot (\text{MPD} - \text{END})}{\text{MPD} + 10}$$

Source code for the HP49G / HP48 series (User RPL language):

```
%%HP: T(3)A(R)F(.);

@ Trimix by blending Heliox and Air, by partial pressure method
@ Using ideal gas laws and metric units
@ Author: Josep Guarro
@ release 1.4 1999-11-25

\<<
@ Data Input
"Equiv. Narc.Depth(m)?" "" INPUT OBJ\->
"Max.Planned Depth(m)?" "" INPUT OBJ\->
"%O2 in Heliox premix?" "" INPUT OBJ\->
"Final pressure (bar)?" "" INPUT OBJ\->

@ Sets local variables with input data
\-> END MPD O2HE PF

@ Core
\<<
21 END 10 + * O2HE MPD END - * + MPD 10 + / @ % O2
MPD END - 100 O2HE - * MPD 10 + / @ % He
21 END 10 + * O2HE MPD END - * + 1000 / @ ppO2
1400 MPD 10 + * 21 END 10 + * O2HE MPD END - * + / 10 - @ MOD 1.4
PF MPD END - * MPD 10 + / @ ppHeOx
PF PF MPD END - * MPD 10 + / - @ ppAir
\>>

@ Sets local variables with computed data
\-> O2 He ppO2 MOD14 ppHeOx ppAir

@ Sets the output display lines
\<<
" Trimix: " O2 0. RND + "/" + He 0. RND +
" O2: " O2 1. RND + "%" +
" He: " He 1. RND + "%" +
" ppO2: " ppO2 2. RND + " bar" +
" MOD: " MOD14 0. RND + " m" +
" Heliox: " ppHeOx 0. RND + " bar" +
" Air: " ppAir 0. RND + " bar" +
\>>

@ Display output
CLLCD
7. DISP
6. DISP
5. DISP
4. DISP
3. DISP
2. DISP
1. DISP
3. FREEZE

@ Clears the stack
CLEAR
\>>
```